
But Our Office Isn’t Paperless—

Can I get all those nifty styles, too?

Or

Using ODS to Create High-quality Hardcopy in SAS v8

[The following is a table, showing the fonts normally available to ODS PRINTER code, that was produced directly
by SAS, with PROC REPORT using the ODS PRINTER destination.  Not exactly a green-bar listing, is it?]



INTRODUCTION
Many of you have seen the nifty styles, highlighting,
and other effects available from the Output Delivery
System, a new feature of the SAS® System in v7.
When we demonstrate this capability, we almost
always show it off using HTML in a web browser.
There are a number of reasons for this—it’s a very
popular technology, it looks good on the computer
screen, we have an embedded browser window in the
Windows version of SAS, and so forth.

But for many of you, it doesn’t solve a lot of the
problems you face in real life, because you need to
produce actual hardcopy or “hardcopy-like” output,
such as PDF or PostScript documents.  This paper
examines alternatives available to you to get the sort
of quality output you need in the formats you require
while still being able to get the “stylish” sorts of
results you’ve seen in the HTML-based demos.

SOLUTIONS
One possible solution, of course, is to simply use the
v7 HTML capabilities to put the output into the
browser, and then use the browser itself to save as
PostScript.  This actually works reasonably well for
small tables, but the browsers do not do a good job
when the table is too long or too wide.

Newly available in production quality for version 8
of the SAS System is the ODS PRINTER output
destination, which allows you produce PostScript or
PCL directly.  In addition, when using the SAS
System under the Windows operating system, ODS
PRINTER code will use the Windows printers
installed on your system.  If you install Adobe
Acrobat (available commercially from Adobe), then
it will produce PDF files by using the PDFWriter
and/or Acrobat Distiller components of Adobe
Acrobat.

In addition, there are three experimental destinations
that may be helpful for producing hardcopy output:
RTF for Microsoft Word, LaTeX and XML
destinations.

Regardless of which of these destinations you
choose, the Output Delivery System uses table
templates and style templates to control the
appearance of the output.

Browsers
As any of you who have used the printing capabilities
of today’s browsers are no doubt aware, they have

some severe limitations when it comes to the
production of hardcopy output.  For example, here is
a rather large table as rendered by Internet Explorer.

Not only are various rules missing, but the titles fail
to repeat on the second page as one would expect.
Even worse, most of the columns of this wide table
are missing entirely.



ODS Printer
New in v8 is the ODS PRINTER destination.
(Actually, it is also available experimentally in v7.)
This destination is designed for printing and produces
results like this (the first and last pages are shown):

  Note that the table is split into panels because it is
wide, and that titles repeat at the top of the next page.

The majority of this paper will focus on examples
using the ODS PRINTER destination.

RTF
Available experimentally in v8 is the RTF
destination.  This is the format used by Microsoft
Word to store documents.  Since it is experimental,
the RTF support cannot be guaranteed always to
work as expected, but we are trying to make it pretty
solid.  The biggest advantage of this format is that it
is relatively easy to add text and/or to modify the
appearance of the output visually.  A small example
of RTF output is shown here.

XML and LaTeX
The XML and LaTeX destinations are also available
experimentally in v8, but are expected to be
somewhat more experimental than RTF.  XML is an
emerging standard for object exchange between
applications.  An increasing number of applications
understand this format, so it will probably be useful
for producing hard copy in the future, but as of this
writing, web browsers and data bases are the primary
applications that support it.

LaTeX is an open-source language-based formatter
for producing “desktop typesetting.”  It is most
popular in academia, particular among
mathematicians. It is in widespread use in the
Applications department at SAS Institute by the folks
working on the mathematically-intensive parts of the
SAS System (SAS/STAT, SAS/OR, SAS/QC, and so
forth).  LaTeX produces high-quality hardcopy
output for a number of printers, but is probably the
solution you are looking for only if you are already
familiar with it.

ODS AND STYLES
F
O

or those who are not familiar with the basics of the
utput Delivery System, following is a brief

ynopsis of ODS, the table templates, and styles.



The Output Delivery System
In previous versions of the SAS System (prior to v7),
each procedure formatted its own output by directly
specifying how to print the data, in much the same
way that you might format the data yourself in a data
step.

With the introduction of the Output Delivery System
in SAS v7, we’ve changed this completely.  Instead
of formatting the text itself, the procedure produces a
data object that describes the data in its full-precision
internal representation—more like a SAS data set
than a listing file—and the name of the template to
use to format the data.

All of the formatting information is contained in the
table template.  Although a default table template is
supplied by the procedure writer, the template is
specified with SAS code and can be modified by the
local site or individual user.

In theory, the table template could supply detailed
information about how to format each aspect of the
individual table columns, but in practice it usually
leaves most of the presentation details to the style
template.  Thus, in GLM, the template specifies that
the “Class” column is in the RowHeader style rather
than specifying that it’s a bold font.

Templates describe the output of most, but not all,
procedures.  Some of the procedures, such PLOT and
GRAPH, produce output which is not tabular at all.
For these procedures, ODS simply reproduces the
listing output in a fixed-size font.  There are also four
procedures which produce tables that are quirky in
some way that makes it impossible to apply the
templates to them.  These are procs PRINT, FREQ
(N-way tables only), TABULATE, and REPORT.

Of these, however, TABULATE and REPORT more
than compensate by having syntax directly in the
procedure syntax that allows you to specify
formatting information for both the table and its
individual components.

The remainder of this paper will use the Table
Templates, special features of procs REPORT and
TABULATE, and the Style Templates to demonstrate
what can be done to produce stylish output intended
for printers.

TABLE TEMPLATES
With a very few exceptions, the tables produced by
the SAS System are described by templates, and the
templates are the means to modify the appearance of
an individual table.  I will use as the example here the

table from PROC STANDARD, simply because it’s
one of the simplest tables in the system.

The Original Table Template
The first step is to determine the original table
template so that we can then proceed to modify it as
we want.  

The first step is to use “ODS TRACE “ to find 
out what the table’s name is:

7?  ods trace on;
8?  proc standard print mean=0; run;

Output Added:
-------------
Name:       Standard
Label:      Standardized data set variables.
Template:   Base.Standard
Path:       Standard.Standard
-------------

Then we get the source code by using PROC
TEMPLATE:

11? proc template;
12? source base.standard;
define table Base.Standard;
   notes "Table definition for PROC
Standard.";
   column name mean std n label;
   define name;
      header = "Name";
      varname = 'Name';
      style = RowHeader;
   end;
   define mean;
      header = "Mean";
      format = D12.;
      varname = 'Mean';
   end;
   define std;
      header = "/Standard/Deviation";
      format = D12.;
      varname = 'stdDev';
   end;
   define n;
      header = "N";
      format = best.;
   end;
   define label;
      header = "Label";
      varname = 'Label';
   end;
   required_space = 3;



   byline;
   wrap;
end;

This produces a table like this:

Modifying the Template
Let us suppose that, for some reason, we wanted to
make the names come out all in italic, and the means
in “reverse video.”  Then we would modify the
template as follows (only the changed parts are
shown below but the entire template must be
supplied):

proc template;
    define table Base.Standard;
       notes "Table definition for PROC
Standard.";
       column name mean std n label;
       define name;
          header = "Name";
          varname = 'Name';
          style =
               RowHeader{font_style=italic};
       end;
       define mean;
          header = "Mean";
          format = D12.;
          varname = 'Mean';
          style = {background=black
                   foreground=white};
       end;
       /* remainder same as before */

Producing a table like this:

PROCS REPORT AND TABULATE
There are two procedures in SAS with special syntax
especially for altering the style.  This example uses
PROC REPORT to produce a font table.  In addition
to showing how this special syntax works, it also
shows the fonts and styles available by default when
using the SAS-supplied PostScript driver, which will
come in handy when you are trying to make stylistic
changes to your own reports.  Here is the code:

title 'ODS Printer Font Table';

data;
  length font_face $100 plain $20
         italic $20 bold $20
         bold_italic $20;
  plain = "plain ABCdef123";
  italic = "italic ABCdef123";
  bold = "bold ABCdef123";
  bold_italic = "bold-italic ABCdef123";
  font_face = "Times"; output;
  font_face = "Helvetica"; output;
  font_face = "Courier"; output;
  font_face = "Symbol"; output;
  font_face = "Bookman"; output;
  font_face = "New Century Schoolbook";
               output;
  font_face = "Palatino"; output;
  font_face = "Chancery"; output;
  font_face = "Dingbats"; output;
  run;

ods printer;

proc report;
  column font_face plain italic bold
         bold_italic;

  define plain /
         style={font_style=roman
                font_weight=medium};
  define italic /
         style={font_style=italic



                font_weight=medium};
  define bold /
         style={font_style=roman
                font_weight=bold};
  define bold_italic /
         style={font_style=italic
                font_weight=bold};

  compute font_face;
    call define(_ROW_, 'STYLE',
      "style={font_face='" ||
              trim(font_face) || "'}");
    endcomp;
run;

ods printer close;

This code first sets up a data set that looks like this:

And then defines columns in PROC REPORT,
setting the attributes of (for example) the “italic”
column to be italic.  For the “font face” column, it
defines the style of the entire row using a call
define statement so that it can use the values from
the data set itself to define the set of fonts to display.
The final table produced is the one shown on the title
page.

OVERALL DOCUMENT STYLE
In addition to the data object supplied by the
procedure and the table template we have already
examined, there is a third element that determines the
appearance of the table: the overall document style,
controlled by a style template.  The style template
comes into play because rather than specifying each
aspect of the presentation directly, the table templates
instead specify the name of a style defined in the
Style Template.  For example, the GLM.ClassLevels
table template contains this line:

define Class;  id header="Class"
                  style=RowHeader;  end;

This means that the appearance of the Class column
will be determined by looking in the Style Template
for the style attributes.  If no style is supplied in the
table template, then the default “Data” style is used.

SAS-supplied styles
The easiest way to change the appearance of your
document is to use on of the styles that SAS supplies
“out of the box.”  To see the list of styles available,
do this:

Proc template; list styles; run;

As of this writing, the styles available are: Default
(actually the HTML default style), Printer (the
default Printer style), Beige, Brown, D3d, Minimal,
Printer, Statdoc, fancyPrinter, sansPrinter,
sasdocPrinter, and serifPrinter.  More may be added
later, and your site may supply local styles as well.

For example, this is some output produced with a
simple “ods printer” statement:

Here is the same output with the specification “ods
printer style=fancyPrinter”

And here again with “ods printer style=brick”,
which is one of the “HTML-oriented” styles:



Color Issues
As you may have noticed, the last example didn’t
really look all that great.  The problem is that it is
using a style designed for HTML browsers, where
color is a given, but it is being reproduced here in
black-and-white (well, grayscale, actually).

This is perhaps the biggest difference between
HTML styles and Printer styles: stylistic distinctions
must be made without the use of color.  (Actually,
this is good advice anyway, for the benefit of any
color-impaired viewers your work might have.)

Moreover, while the printers theoretically support
“grayscale,” they all in fact dither the grayscale out
of little dots.  This works just fine for filled
backgrounds, and in the default printer style, but it
makes narrow rules look more dotted than gray.
Similarly, all but the largest fonts look terrible when
set in any color other than black on a typical 300 dpi
office printer.  So all the examples in the remainder
of this paper will avoid using colors with differing
hues entirely, and will use even grayscale only for
backgrounds.

If you are lucky enough to have a 600 dpi or higher,
or a color printer available, you will have more
freedom.  Indeed, if you have color available, you
can use all of the HTML-oriented styles can get
perfectly pleasing results.

Creating your own style
It is possible to create your own style, and customize
to your heart’s content.  The “base” style for the
entire system is called “default,” and you can print it
out by using

Proc template; source styles.default; run;

However, it is enormous and only rarely will you
need to deal with all the elements in that style.  Most
styles simply want to alter the colors and/or fonts
used by the style, and perhaps change one or two
individual elements.  Conveniently, the default ODS
PRINTER style, “styles.printer”, does exactly this,

and is a template of much more manageable size, so
for most purposes you will probably do this:

Proc template; source styles.printer; run;

To see the default printer style, and use that as a
starting point for your own styles.  If you wish to
change the appearance of an individual element of
the document, such as the page number, the date,
system titles, or all data cells, though, you will need
to use the “styles.default” as your starting point for
altering those elements.  For example, here is a
document style in which we start with elements
copied and modified from “styles.printer”, and then
add the TitlesAndFooters element from styles.default,
so that we get  this code:
proc template;
define style Styles.mystyle;
   parent = styles.Printer;

   /*--- from styles.Printer ---*/
   replace fonts /
      'TitleFont2' =
         ("Helvetica",12pt,Bold)
      'TitleFont' =
         ("Helvetica",13pt,Bold)
      'StrongFont' =
         ("Times",10pt,Bold)
      'EmphasisFont' =
         ("Times",10pt,Italic)
      'FixedEmphasisFont' =
         ("Courier",9pt,Italic)
      'FixedStrongFont' =
         ("Courier",9pt,Bold)
      'FixedHeadingFont' =
         ("Courier",9pt,Bold)
      'BatchFixedFont' =
         ("Courier",7pt)
      'FixedFont' =
         ("Courier",9pt)
      'headingEmphasisFont' =
         ("Chancery",13pt,Bold Italic)
      'headingFont' =
         ("Chancery",11pt,Bold)
      'docFont' =
         ("Times",10pt,italic);
   style Table from Output /
      rules = ROWS
      frame = BOX
      cellpadding = 4pt
      cellspacing = 0.50pt
      borderwidth = 3pt
      BorderColor = gray99
      Background = _undef_;
   replace color_list
      'bgH' = white
      'fg' = black



      'bg' = white;

    /*--- from styles.default ---*/
    replace TitlesAndFooters
            from Container /
      background = gray33
      foreground = white;

end;
run;

ods printer style=mystyle file="p7.ps";
proc standard print; run;
ods printer close;

A couple subtleties should be noted here.  First, in the
Table definition, the “background=_undef_” is
necessary; the default table background is black,
which effectively forces rules on everywhere, since
the table background color “shows through” in the
spacing between the table cells.  Second, it is
necessary to completely re-define any element if you
want to replace part of it; thus, we had to list all of
the fonts even though we were only changing a few
of them.

This is the result:

Admittedly, this is not likely to win any design
awards, but it does show the significant changes that
can be made with styles.

MODIFYING HTML STYLE EXAMPLES
If you have seen other examples of “stylish” SAS
output which uses HTML, there are a few points that
you need to be aware of it you want to get nice-
looking printer output:

• As previously mentioned, colors are no available
on most office printers.  Some colors happen to

work well when translated to grayscale; others
do not.

• On some printers, even the grayscale which ODS
PRINTER code uses by default does not look
very good.  This is particularly true if you are
going to turn around and FAX or photocopy the
results.  If this is the case for your printer, use
the NOCOLOR option for ODS PRINTER and it
will not draw any of the background colors.

• ODS PRINTER does not support images in v8.
This feature will be supported for v9.  If you
must have images in printed output, you should
use HTML and print from the browser.  Of
course, that won’t work well for large tables, but
I doubt that there are very many large tables with
images that need to be printed.  (I hope not,
anyway.)

The following take a couple of the examples from the
“Your Base SAS Dollars At Work” paper and re-
work them for the printer.  All the examples that
modify table templates to alter the arrangement of
data, or that show off new features of the procedures,
of course, work just fine as they stand.

PROC REPORT “traffic-lighting”
For example, here is a PROC REPORT example
done in an HTML style.  Only part of the table is
shown in order to make the results visible.

The style elements can be changed to rely less on
color changes and more on other distinctions; for
example, take the original specification
style(COLUMN)={foreground=yellow
               font_weight=bold};

and change it into this:

style(COLUMN)=
         {font=("New century schoolbook",
                10pt, bold italic)};

Similarly, take the specification for the missing data
and change the pink background into this:

call define('COMMENTS',"STYLE",



             "style(CALLDEF)={just=c
                  fillrulewidth=0.5pt}");

(The fillrulewidth specification causes the line to be
filled in with a rule.)  The result looks like this:

Of course, in practice you may find that the HTML
styles only need a little tweaking to work just fine for
your printing purposes.  For example, in converting
the HTML to grayscale for the screen, everything but
the yellow actually comes out just fine.  It is
important, however, to try looking at the results on a
printer before assuming that the styles that work well
for HTML will work well for PostScript as well.

A PROC TABULATE example
Similarly, the PROC TABULATE example from the
“Base Dollars” paper has this specification:

proc tabulate s={foreground=green};
class region citysize saletype
        / s={foreground=blue};
   classlev region citysize saletype
        / s={foreground=yellow};
   var quantity amount
        / s={foreground=black};
   keyword all sum / s={foreground=white};
   format saletype $salefmt.;
   label region="Region" citysize="Citysize"
                        saletype="Saletype";
   label quantity="Quantity"
         amount="Amount";
         keylabel all="Total";

   table all={label = "All Products"
              s={foreground=orange}},
            (region all)*(citysize
           all*{s={foreground=#002288}}),
          (saletype all)*(quantity*f=COMMA6.
                       amount*f=dollar10.)
          /  s={background=red}
           misstext={label="Missing"
                     s={foreground=brown}}
           box={label="Region by Citysize by
Saletype" s={foreground=purple}};

which produces a table like this:

By changing the colors into font specifications, we
can transform it into a form where the same
distinctions can be seen even in the extreme case
where no grayscales can be used at all.  Here is the
new tabulate code:

proc tabulate s={font_face=bookman};
   class region citysize saletype
         / s={font_style=italic};
   classlev region citysize saletype
      / s={font_style=italic
           font_weight=bold};
   var quantity amount
       / s={font_weight=bold};
   keyword all sum
         / s={font=("helvetica",
                    10pt, bold italic)};
   format saletype $salefmt.;
   label region="Region" citysize="Citysize"
         saletype="Saletype";
   label quantity="Quantity"
         amount="Amount";
   keylabel all="Total";

   table all={label = "All Products"
              s={font_face=chancery}},
            (region all)*(citysize
            all*{s={font_face=helvetica
                font_weight=bold}}),
          (saletype all)*(quantity*f=COMMA6.
                      amount*f=dollar10.)
          / s={fillrulewidth=1pt}
           misstext={label="Missing"
           s={font_face=courier
              font_style=italic}}
           box={label="Region by Citysize by
Saletype"   s={font_face=helvetica
               font_style=italic}};
   run;

Which produces a table like this (in NOCOLOR
mode):



CONCLUSION
There are a number of ways to control and alter the
appearance of the SAS output for all of the ODS
destinations other than LISTING and OUTPUT.
Most of the stylistic elements will work for any
destination, but the best-looking results will be
obtained if the style used is sensitive to the intended
destination. This paper has only scratched the surface
of how to best get “stylish” output for printers from
the SAS system, and we hope that this topic will 
be more extensively explored as more people 
get familiar with the new ODS capabilities.




